

```

// Omniscan macro for use with WordPerfect for Windows

// Tell Word Perfect that we're using Word Perfect
Application(wp;wpwp;default;"wpwpus.wcd")

// Define the temporary file that we'll be using
Assign(tempfile;"c:\ocrtemp.wp")

// Load in the DLL and save the returned id
DLLLoad(DllId;"omniscan.dll")

// Initiate a link to the OCR Server; if failed, bail
MacroStatusPrompt(On!;"Initiating Link to OCR Server")
DllCall(DllId;"OcrInitiateLink";status:WORD;{})
If (status <> 0)
    Prompt("Error";"Failed to initiate link to OCR Server";;;)
    Pause
    Go (Bail@)
Endif

// Perform the actual OCR. Since WordPerfect treats an integer
// return value as UNsigned, the -11 code returned when the user
// cancels is actually thought to be a value of 65525. Hence, the
// test below must check for 65525 instead of -11.
MacroStatusPrompt(On!;"Performing OCR")
DllCall(DllId;"OcrPerform";status:WORD;{tempfile;"WordPerfect 5.1"})
If ((status <> 0) And (status <> 65525))
    Prompt("Error";"Failed to complete OCR";;;)
    Pause
Endif

// Close the link to the OCR Server
MacroStatusPrompt(On!;"Closing Link to OCR Server")
DllCall(DllId;"OcrCloseLink";status:WORD;{})
If (status <> 0)
    Prompt("Error";"Failed to close link to OCR Server";;;)
    Pause
Endif

// If you have just opened a "new" document and never entered or
// deleted any text to/from it, then WordPerfect will treat the
// "FileRetrieve" with insert command below as a file open.
// Otherwise, the file is properly inserted. Since the file is
// deleted when we get done, we need to make sure the current
// document is considered NOT to be "new".
GetWPData(data;DocumentBlank!)

```

```
If (data = TRUE)
    Type(" ")
    DeleteCharPrevious()
Endif

// Open and insert the temporary file into the current document,
// then delete it
DllCall(DllId;"OcrFileCheck";status:WORD;{tempfile})
If (status = 0)
    FileRetrieve(tempfile;Yes!;Insert!)
    DllCall(DllId;"OcrDeleteFile";status:WORD;{tempfile})
Endif

// Restore the prompt line to normal and unload the DLL
Label(Bail@)
MacroStatusPrompt(Off!;"")
DLLFree(DllId)
```